



Awan Berarak

**Membuat Notasi Rumus Jumlah
Terbilang / Terjemahan Uang
pada Microsoft Excel**



ratusan

Ratusan juta

Ratusan milyar

11/16/14

Terjemahan Uang Pada Microsoft Excel tanpa Visual Basic

Kini dari waktu ke waktu, tidak terasa sudah lima hari kita menjalani waktu di tahun 2013 ini. Banyak hal yang kita alami, tetapi suatu kreatifitas tidak akan pernah berhenti. Harapan itu adalah semangat, berfikir kreatif akan lebih bersemangat lagi.

Kali ini saya sedang berbagi pengetahuan dan pengalaman dengan anda sahabat saya yang tercinta, masih pada topik Microsoft Excel.

Ada juga yang menyebut sebagai jumlah terbilang atau terbilang. Tetapi saya mengambil judul Terjemahan Uang Pada Microsoft Excel tanpa Visual Basic. Mungkin judul yang tepat karena memang hanya mengandalkan fungsi yang ada di Microsoft Excel saja tanpa ada unsur software atau aplikasi dari luar Microsoft Excel.

Kita langsung saja :


Untuk mendapatkan terjemahan uang pada Microsoft Excel kita menggunakan beberapa fungsi antara lain :

- fungsi IF
- fungsi INT
- fungsi SUM
- fungsi VLOOKUP

Dengan fungsi-fungsi diatas, rasanya sudah cukup untuk membuat sebuah kwitansi sehingga pada kwitansi jika diketikkan nilai nominal misalnya sebesar Rp. **2.350.753,00** maka pada bagian penjelasan dengan huruf maka akan langsung diterjemahkan menjadi "Dua juta tiga ratus lima puluh ribu tujuh ratus lima puluh tiga rupiah", tanpa harus mengetikkan nilai tersebut dengan kalimat.

Nah bagaimana caranya ?

Karena ini proses yang memang sedikit memerlukan konsentrasi, maka saya akan bagi dalam beberapa posting . Untuk yang pertama ini saya akan memulai dengan nilai ratusan terlebih dahulu.



Sesuai dengan gambar di atas, nilai ratusannya adalah 753, untuk itu perlu diuraikan terlebih dahulu.

Jika Microsoft Excel anda sudah aktif, terlebih dahulu yang perlu disiapkan adalah daftar untuk terjemahan. Bisa pada Sheet1 bisa juga pada Sheet2. Pada file contoh yang dapat anda [Download](#), saya menggunakan Sheet2 yang saya ganti namanya menjadi Huruf (ini nama sheet), seperti gambar berikut :

	A	B	C
1			
2		NILAI	HURUF
3		0	
4		1 satu	
5		2 dua	
6		3 tiga	
7		4 empat	
8		5 lima	
9		6 enam	
10		7 tujuh	
11		8 delapan	
12		9 sembilan	
13		10 sepuluh	
14		11 sebelas	
15		12 dua belas	
16		13 tiga belas	
17		14 empat belas	
18		15 lima belas	
19		16 enam belas	
20		17 tujuh belas	
21		18 delapan belas	
22		19 sembilan belas	
23		20 dua puluh	
24		100 seratus	
25		1000 seribu	

Pada contoh, saya memulai pada Cell B3 sampai dengan Cell 25 yang nantinya menggunakan fungsi VLOOKUP untuk menampilkan terjemahannya.

Jika daftar ini sudah anda siapkan, maka kita mulai dengan nilai ratusan terlebih dahulu, yaitu 3 digit terakhir.

Kita kembali pada Sheet1,

pada Cell B2 ketikkan angka 753.

pada Cell C2 input notasi rumus ini : **=B2-**

(INT(B2/1000)*1000).

Tahap-tahapan notasi rumus tersebut adalah :

1. $=C2/1000$ hasilnya adalah 0.753
2. Tambahkan fungsi INT didepannya sehingga menjadi $=INT(C2/1000)$ hasilnya adalah 0 (nol)
3. Hasil yang kedua kita kalikan lagi dengan 1000 sehingga notasinya menjadi $=(INT(C2/1000)*1000)$ dan hasilnya adalah 0 (nol) juga. **Bingung?**
4. Terakhir, agar hasilnya menjadi murni ratusan maka Cell B2 kita kurangi dengan hasil yang ke 3 diatas sehingga notasi rumusnya adalah **$=B2-(INT(C2/1000)*1000)$** . Nah, disini hasilnya menjadi 753.

Coba..! pada Cell B2 anda input dengan nilai misalnya 350753, maka hasil pada Cell C2 tetap 753. Dan dari sinilah kita akan menterjemahkan uangnya menjadi huruf dengan fungsi VLOOKUP dari tabel yang anda buat pada sheet2 tadi.

Kita lanjutkan lagi yah..... (minum air putih banyak-banyak biar nggak tegang dan sekali-kali lihat keluar jendela, nikmati keindahan yang disuguhkan oleh Allah SWT di luar sana).

Tahap berikutnya adalah :

1. Pada Cell C4 input fungsi rumus ini : **$=C2-(INT(C2/10)*10)$** hasilnya adalah 3 (tiga)
2. Pada Cell C5 input fungsi rumus ini : **$=(INT(C2/10)/10)*10-(C6*10)$** hasilnya adalah 5 (lima)
3. Pada Cell C6 input fungsi rumus ini : **$=INT(C2/100)$** hasilnya adalah 7 (tujuh)

Jangan bingung ya...., mari konsentrasi penuh

1. Pada Cell E4 input fungsi rumus ini :
 $=IF(C4+C5+C6=0,"",IF(C5=1,VLOOKUP((C5*10)+C4,HURUF!B3:K43,2),VLOOKUP(C4,HURUF!B3:K43,2)))&" rupiah"$ hasilnya adalah **tiga rupiah**
2. Pada Cell E5 input fungsi rumus ini
 $:=IF(C5=0,"",IF(C5=1,"",VLOOKUP(C5,TERJEMAHAN,2)&" puluh"))$
hasilnya adalah **lima puluh**
3. Pada Cell E4 input fungsi rumus ini
 $:=IF(C6=0,"",(IF(C6=1,VLOOKUP(100,TERJEMAHAN,2),VLOOKUP(C6,TERJEMAHAN,2)&" ratus")))$ hasilnya adalah **tujuh ratus**

Keterangan : yang berwarna merah adalah rumus dan yang berwarna biru adalah hasil terjemahan dalam bentuk kalimat (huruf).

	A	B	C	D	E
1		input			
2		9,122,350,753	753		
3			3	tiga rupiah	
4		satuan sampai ratus	5	lima puluh	
5			7	tujuh ratus	
6					
7					

<http://jempot2-office.blogspot.com/>

Hasil terjemahan

satuan

Belasan dan puluhan

ratusan

Sekarang kita masuk pada tahap penggabungan hasil perumusan terjemahan tersebut dengan cara menginput notasi rumus ini : **=E6&" "&E5&" "&E4** pada Cell G2. Hasilnya adalah **tujuh ratus lima puluh tiga rupiah** dalam huruf kecil.

Jika ingin seperti ini Tujuh Ratus Lima Puluh Tiga Rupiah maka rumusnya ditambah dengan PROPER menjadi **=PROPER(E6&" "&E5&" "&E4)**, atau jika ingin huruf kapital semua maka ditambahkan dengan UPPER menjadi : **=UPPER(E6&" "&E5&" "&E4)**, jika ingin huruf kecil semua tambahkan dengan LOWER didepan notasi rumus tersebut dan ini juga tergantung pada penulisan listing terjemahan pada sheet2 tadi. Tetapi sobat dapat mengaturnya dengan penambahan fungsi seperti tersebut diatas.

Lumayan panjang juga prosesnya, apalagi diuraikan fungsi VLOOKUPnya, waduh..., satu posting tak akan kelar 1 hari. Saya tidak janji, semoga pada posting berikutnya saya uraikan maksud dari notasi tersebut. Semoga dan semoga pula teman-teman yang membaca artikel ini mendapatkan manfaat.

Amiiin....

Terjemahan Uang Pada Microsoft Excel : Ratusan Ribu

Pada posting kali ini, kita sampai pada tahap ratusan ribu untuk penterjemahan uang pada Microsoft Excel. Setelah pada posting saya tanggal 5 Januari 2013 dengan titel [Terjemahan Uang Pada Microsoft Excel tanpa Visual Basic](#) yang membahas terjemahan uang dalam ratusan.

Langsung dimulai :

Seperti pada contoh angka : Rp. 2.350.753,00 maka untuk mendapatkan angka dalam ratusan ribu maka beberapa notasi diperlukan diantaranya adalah :

Pada Cell C8 input notasi ini : **=INT(B2/1000)-INT((INT(B2/1000))/1000)*1000** hasilnya adalah **350**.

Selanjutnya perlu diuraikan satu persatu, maka berikutnya adalah

Pada Cell C10 input notasi ini : **=C8-(INT(C8/10)*10)** hasilnya adalah **0** (nol)

Pada Cell C11 input notasi ini : **=(INT(C8/10)/10)*10-(C12*10)** hasilnya adalah **5** (lima)

B8			f _x	ratusan ribu		
	A	B	C	D	E	
1		input				
2		3,572,350,753	753			
3						
4		satuan sampai ratus	3		tiga rupiah	
5			5		lima puluh	
6			7		tujuh ratus	
7						
8		ratusan ribu	350			
9		HTTP://JEMPOT2-OFFICE.BLOGSPOT.COM/				
10			0		ribu	
11			5		lima puluh	
12			3		tiga ratus	
13						

Pada Cell C12 input notasi ini : **=INT(C8/100)** hasilnya adalah **3** (tiga)

Tahap berikutnya adalah mengambil data lookup dari sheet2 dengan fungsi vlookup.

1. Pada Cell E10 input notasi ini :
`=IF(C10+C11+C12=0,"",IF(C11=1,VLOOKUP((C11*10)+C10,HURUF!TERJEMAHAN,2),VLOOKUP(C10,HURUF!TERJEMAHAN,2))&" ribu")` hasilnya adalah **ribu**
2. Pada Cell E11 input notasi ini :
`=IF(C11=0,"",IF(C11=1,"",VLOOKUP(C11,HURUF!TERJEMAHAN,2)&" puluh"))` hasilnya adalah **lima puluh**
3. Pada Cell E12 input notasi ini :
`=IF(C12=0,"",IF(C12=1,VLOOKUP(100,HURUF!TERJEMAHAN,2),VLOOKUP(C12,HURUF!TERJEMAHAN,2)&IF(C10=0," ratus",IF(C10>=1," ratus"," ratus ribu"))))` dan hasilnya adalah **tiga ratus**.

07 January 2013

[Home](#) > [Terjemahan uang pada Microsoft Excel](#) > Terjemahan Uang Pada Microsoft Excel : Ratusan Juta

Terjemahan Uang Pada Microsoft Excel : Ratusan Juta

Jika pada posting saya yang berjudul [Terjemahan Uang Pada Microsoft Excel : Ratusan Ribu](#) , maka kali ini kita sampai pada cara menterjemahkan uang dengan Microsoft Excel pada tahap ratusan juta.

Jadi seperti pada judul posting Terjemahan Uang Pada Microsoft Excel tanpa Visual Basic , kenyataannya memang Visual Basic belum dibutuhkan disini, apalagi ini hanya untuk urusan yang tidak begitu besar, dan ini ditujukan bagi teman-teman yang ingin independen dalam bekerja.

Oke..., mari kita lanjutkan

Seperti contoh yang lalu hanya berkisar pada dua jutaan, maka contoh tersebut kita kembangkan menjadi ratusan juta. Kita buat contoh dengan angka Rp. 972.350.753,00

Tahap-tahapnya sama dengan posting yang lalu, tetapi notasi rumusnya tentu berbeda.

Siap....?

Letakkan cursor pada Cell C14 dan input notasi fungsi ini :

`=INT((INT(B2/1000))/1000)-(INT((INT(B2/1000))/1000000)*1000)` hasilnya adalah **972**

Berikutnya adalah :

Pada Cell C16 input notasi rumus ini : `=C14-(INT(C14/10)*10)` hasilnya adalah **2** (dua)

Pada Cell C17 input notasi rumus ini : `=(INT(C14/10)/10)*10-(C18*10)` hasilnya adalah **7** (tujuh)

Pada Cell C18 input notasi rumus ini : `=INT(C14/100)` hasilnya adalah **9** (sembilan)

Selanjutnya kita akan kembali ambil terjemahan hurup dari Sheet2 (yang di rename menjadi HURUF) untuk menterjemahkan nilai-nilai mulai dari Cell C16, Cell C17 dan Cell C18.

Begini ini rumusnya :

1. Pada Cell E16 isikan dengan notasi rumus ini :
`=IF(C16+C17+C18=0,"",IF(C17=1,VLOOKUP((C17*10)+C16,HURUF!TERJEMAHAN,2),VLOOKUP(C16,HURUF!TERJEMAHAN,2))&" juta")` hasilnya adalah **dua juta**

2. Pada Cell E17 isikan dengan notasi rumus ini :
`=IF(C17=0,"",IF(C17=1,"",VLOOKUP(C17,HURUF!TERJEMAHAN,2)&"puluh"))` hasilnya adalah **tujuh puluh**
3. Pada Cell E18 isikan dengan notasi rumus ini :
`=IF(C18=0,"",IF(C18=1,VLOOKUP(100,HURUF!TERJEMAHAN,2),VLOOKUP(C18,HURUF!TERJEMAHAN,2)&" ratus"))` dan hasilnya adalah **sembilan ratus**

Berhubung ruangan posting ini jadi sempit karena rumusnya panjang, maka ada yang hurufnya dikecilkan. Jadi Jempot2-office mohon maaf. He...he...he....

Untuk menggabungkan terjemahan tersebut masukkan notasi rumus ini pada Cell G2 :
`=PROPER(E24&" "&E23&" "&E22&" "&E18&" "&E17&" "&E16&" "&E12&" "&E11&" "&E10&" "&E6&" "&E5&" "&E4)` hasilnya adalah : **Sembilan Ratus Tujuh Puluh Dua Juta Tiga Ratus Lima Puluh Ribu Tujuh Ratus Lima Puluh Tiga Rupiah**

	A	B	C	D	E
1		input			
2		972,350,753	753		
3					
4		satuan sampai ratus	3		tiga rupiah
5			5		lima puluh
6			7		tujuh ratus
7					
8		ratusan ribu	350		
9		HTTP://JEMPOT2-OFFICE.BLOGSPOT.COM/			
10			0		ribu
11			5		lima puluh
12			3		tiga ratus
13					
14		ratusan juta	972		
15					
16			2		dua juta
17			7		tujuh puluh
18			9		sembilan ratus
19					

Tahap berikutnya adalah ratusan milyar, lebih jelasnya silakan download file contoh, tapi dapat pula langsung dipergunakan. he...he..., semoga anda mendapatkan Rahmat dan Hidayah dari Allah SWT.

Amiinnn.....

Terjemahan Uang Pada Microsoft Excel : Ratusan Milyar

Pada prinsipnya untuk Terjemahan Uang Pada Microsoft Excel : Ratusan Milyar ini tidak berbeda dengan cara-cara yang lain yang sudah saya posting pada posting yang sudah lewat (lihat : [Terjemahan Uang Pada Microsoft Excel tanpa Visual Basic](#) , [Terjemahan Uang Pada Microsoft Excel : Ratusan Ribu](#) dan [Terjemahan Uang Pada Microsoft Excel : Ratusan Juta](#)).

Untuk itu tidak berpanjang lebar kita langsung pada topik yang akan dibahas.

http://jempot2-office.blogspot.com/

	A	B	C	D	E	F
1		input				
2		948,972,350,753	753			
3						
4		satuan sampai ratus	3		tiga rupiah	
5			5		lima puluh	
6			7		tujuh ratus	
7						
8		ratusan ribu	350			
9						
10			0		ribu	
11			5		lima puluh	
12			3		tiga ratus	
13						
14		ratusan juta	972			
15						
16			2		dua juta	
17			7		tujuh puluh	
18			9		sembilan ratus	
19						
20		ratusan milyar	948			
21						
22			8		delapan milyar	
23			4		empat puluh	
24			9		sembilan ratus	
25						

Pada Cell C20 pasang notasi rumus ini :

=INT((B2/1000000000)) hasilnya adalah **948**

Pada Cell C22 pasang dengan rumus : **=C20-(INT(C20/10)*10)** hasilnya adalah **8** (delapan)

Pada Cell C23 pasang dengan rumus : **=(INT(C20/10)/10)*10-(C24*10)** hasilnya adalah **4** (empat)

Pada Cell C24 pasang dengan rumus : **=INT(C20/100)** hasilnya adalah **9** (sembilan)

Pengambilan data dari Sheet2 sebagai berikut :

1. Pada Cell E22 pasang dengan rumus :
`=IF(C22+C23+C24=0,"",IF(C23=1,VLOOKUP((C23*10)+C22,HURUF!TERJEMAHAN,2),VLOOKUP(C22,HURUF!TERJEMAHAN,2))&" milyar")` hasilnya adalah **delapan milyar**
2. Pada Cell E23 pasang dengan rumus :
`=IF(C23=0,"",IF(C23=1,"",VLOOKUP(C23,HURUF!TERJEMAHAN,2)&" puluh"))` hasilnya adalah **empat puluh**
3. Pada Cell E24 pasang dengan rumus :
`=IF(C24=0,"",IF(C24=1,VLOOKUP(100,HURUF!TERJEMAHAN,2),VLOOKUP(C24,HURUF!TERJEMAHAN,2)&" ratus"))` hasilnya adalah **sembilan ratus**

Sampai disini proses pembuatan terjemahan uang pada Microsoft Excel sampai dengan nilai milyaran sudah rampung dan perlu menggabungkan dari masing-masing bagian sehingga menjadi suatu kalimat yang lengkap. Untuk itu pada Cell G2 masukkan notasi rumus ini :
`=PROPER(E24&" "&E23&" "&E22&" "&E18&" "&E17&" "&E16&" "&E12&" "&E11&" "&E10&" "&E6&" "&E5&" "&E4)` dan hasilnya adalah **Sembilan Ratus Empat Puluh Delapan Milyar Sembilan Ratus Tujuh Puluh Dua Juta Tiga Ratus Lima Puluh Ribu Tujuh Ratus Lima Puluh Tiga Rupiah.**

Anda dapat menggunakan terjemahan uang ini untuk keperluan seperti kwitansi atau apa saja yang dibutuhkan dalam membuat kalimat terjemahan dari suatu nilai nominal.
 Selamat mencoba.

